
Bayesian Multivariate Prior for Multiple Linear
Regression

Overview

This example uses the MCMC procedure to fit a Bayesian multiple linear regression (MLR) model by using
a multivariate prior on the regression parameters. It demonstrates how to use existing SAS multivariate
density functions for specifying prior distributions.

The SAS source code for this example is available as a text file attachment. In Adobe Acrobat, right-click
the icon in the margin and select Save Embedded File to Disk. You can also double-click the icon to open
the file immediately.

Analysis

Researchers are interested in determining the relationship of gestational length and litter size on brain weight
after accounting for body weight. They study the evalutionary benefits of larger brain weight after account-
ing for the coinciding disadvantages of longer gestational length and smaller litter size.

The following data set contains the average brain and body weights in grams and kilograms, respectively,
the average gestational length in days, and the litter size for 95 different species as given in Sacher and
Staffeldt (1974):

data brainweight;
input body gestation litter @@;
log_brain = log(brain);
log_body = log(body);
log_gestation = log(gestation);
datalines;
17.5 3.5 26 1
3.5 0.93 34 4.6
3.15 0.15 46 3
1.14 0.049 51 1.5

... more lines ...

200 39 180 1
210 66 158 1.2
125 49 150 2.4
106 30 151 2



/*-----------------------------------------------------------------
  Example: Fitting a Bayesian Linear Regression model with a 
           Multivariate Prior     
  Requires: SAS/STAT
  Version: 9.2                                                                       
  ------------------------------------------------------------------*/

data brainweight;
   input body gestation litter @@;
   log_brain = log(brain); 
   log_body = log(body);
   log_gestation = log(gestation); 
   datalines;
   17.5 3.5 26 1
   3.5 0.93 34 4.6
   3.15 0.15 46 3
   1.14 0.049 51 1.5
   1.37 0.064 46 1.5
   22 2.1 135 1
   12.8 1.2 90 1.2
   9.9 0.7 135 1
   54 7.7 139 1
   73 3.7 180 1
   114 9.1 140 1
   109 7.7 140 1
   7.8 0.22 145 2
   84.6 6 175 1
   107 8.7 165 1.1
   183 21 180 1
   179 32 180 1
   67 4.6 195 1
   65.5 5.8 168 1
   102 5.5 210 1
   343 37 270 1
   360 45 230 1
   406 140 265 1
   1300 65 270 1
   12 3.7 120 4
   9.6 2.2 31 5
   13.3 2.9 41 2.5
   6.23 0.33 38 3
   1.89 0.052 40 3.1
   40 20 128 2.9
   45 25 128 4
   0.68 0.027 23 3.7
   0.63 0.026 23 5
   0.52 0.017 24 5
   0.69 0.024 24 5
   0.67 0.036 21 4.6
   1.12 0.13 16 6.3
   1.04 0.065 21 4
   0.72 0.05 23 7.3
   2.38 0.34 21 8
   0.45 0.024 19 5
   1.18 0.15 27 5.6
   37 14 112 1.2
   24 6.6 113 1
   4.28 0.97 67 2.6
   76 30 123 3
   20.3 2.8 104 1.3
   9.9 0.78 98 1.2
   5.25 0.43 110 2
   23 5 132 5.5
   1600 160 360 1
   537 56 270 1
   70.2 8.5 63 4
   48 6 52 4
   37.3 3.8 63 3.7
   28.5 3.2 65 4
   400 250 219 2.3
   500 250 240 1.8
   41.6 5.3 63 3.5
   31.2 2 77 1.1
   53 6 60 2.2
   28.4 2.5 63 4
   75 12 60 2.5
   157 46 92 2.5
   260 180 108 3
   302 210 104 3
   355 250 254 1
   363 100 343 1
   442 110 240 1
   550 400 310 1
   20.5 3.8 225 2.4
   712 480 330 1
   250 230 390 1
   185 150 120 4
   180 190 115 8
   590 1400 240 1
   260 150 205 1
   225 93 330 1
   198 45 300 1.1
   124 16 183 1.1
   223 80 240 1
   219 89 218 1
   435 200 255 1
   365 120 235 1
   383 120 246 1.1
   288 110 225 1
   480 560 255 1
   334 250 255 1
   456 520 280 1
   93 13 120 1
   200 39 180 1
   210 66 158 1.2
   125 49 150 2.4
   106 30 151 2
   ;

proc template;
   define statgraph brainweight;
      begingraph;

     layout lattice / rows=2 columns=2;

    layout overlay;
         histogram brain;
      endlayout;

    layout overlay;
         histogram body;
      endlayout;

    layout overlay;
         histogram gestation;
      endlayout;

    layout overlay;
         histogram litter;
      endlayout;

     endlayout;
     endgraph;
end;
run;

ods graphics on;
proc sgrender data=brainweight template=brainweight;
run;
ods graphics off;

ods graphics on;
proc mcmc data=brainweight nbi=5000 nmc=25000 thin=5 seed=1181
   propcov=quanew monitor=(_parms_ ExpBeta2 ExpBeta3);
   
   array mu_pr[4] mu_pr0-mu_pr3;
   array beta[4] beta0-beta3;
   array data[4] 1 log_body log_gestation litter;

   begincnst;
      call zeromatrix(mu_pr);     
      rc = logmpdfset('lp',16,0,16,0,3,2.25,0,-2,0,1);
   endcnst;

   parms beta: 0;                                 
   parms sig2 1;
   
   ExpBeta2=exp(beta2);
   ExpBeta3=exp(beta3);

   lmvn = logmpdfnormal(of beta0-beta3, of mu_pr0-mu_pr3, 'lp');
   prior beta: ~ general(lmvn);
   prior sig2 ~ igamma(shape = 2.001, scale = 1.001);

   call mult(beta, data, mu);
   model log_brain ~ n(mu, var = sig2);
run;

data _null_;
   rc = logmpdffree();
   put rc=;
run;
ods graphics off;  


SAS source code for this example. Right-click to save file.

;

Figure 8 displays the distribution of each variable. The skewed distributions and the differing orders of
magnitude for the minimum and maximum brain weight, body weight, and gestational length suggest the
need for the log transformation.

Figure 8 Distributions of Variables

Bayesian MLR Model

Suppose you want to fit a Bayesian MLR model for the logarithm of brain weight with density as

log.BRAIN/i � normal.�i ; �2/

�i D Xiˇ (11)

where Xi is the vector of covariates listed as Xi D f1 log.BODY/i log.GESTATION/i LITTERig for
i D 1; :::; n species. The African elephant has been omitted from this example because it is easily recognized
as an extreme outlier even on the log scale.

The likelihood function for the logarithm of the brain weight and corresponding covariates is

p.log.BRAIN/i jXi ;ˇ/ D normal.�i ; �2/ (12)

where p.�j�/ denotes a conditional probability density. The normal density is evaluated at the specified value
of log.BRAIN/i and the corresponding mean parameter �i , which is defined in Equation 11. The four
regression parameters in the likelihood are ˇ0 through ˇ3.

Suppose you had expert or prior knowledge that some of the covariates were correlated. You might want
to use a multivariate prior to incorporate your information. Using the multivariate normal prior, you enable
covariates to be independent or correlated a priori. You can also specify the a priori correlation that you
believe to be positive or negative. Suppose you thought, a priori, that body weight and gestational age were
positively correlated and that body weight and litter size were negatively correlated. Calculate the prior
covariance as the product of the prior correlation and standard deviations. More formally, the formula for
calculating the covariance of the j and k covariate is �jk D �jksj sk where �jk is the prior correlation and
sj and sk are the sample standard deviations of the j th and kth covariates, respectively.

Suppose the prior correlation between the first and second covariate was �12 D 0:5, and a priori you thought
the standard deviations of the regression coefficients for log of body weight and gestational age were s1 D 4
and s2 D 1:5, respectively. The prior covariance of the log of body weight and gestational age can be
calculated as �12 D �12s1s2 D .0:5/.4/.1:5/ D 3. Similarly, suppose the prior correlation between the
first and third covariate was �13 D �0:5 and a priori the standard deviation of the litter size was s3 D 1.
The prior covariance of the log of body weight and litter size can be calculated as �13 D �13s1s3 D

.�0:5/.4/.1/ D �2.

Suppose the following prior distributions are placed on the parameters:

�.ˇ/ D MVN4.�pr D 0; †/ (13)

† D

2664
16 0 0 0

0 16 3 �2

0 3 2:25 0

0 �2 0 1

3775
�.�2/ D igamma.shape D 2:001; scale D 1:001/ (14)

where �.�/ indicates a prior distribution, MVN4.�pr ; †/ indicates a multivariate normal prior of dimension
four with mean vector �pr and variance matrix†, and igamma indicates the density function for the inverse-
gamma distribution.

The following SAS statements use the likelihood function and prior distributions to fit the Bayesian MLR
model with the multivariate prior. The PROC MCMC statement invokes the procedure and specifies the
input data set. The NBI= option specifies the number of burn-in iterations. The NMC= option specifies the
number of posterior simulation iterations. The THIN=5 option specifies that one of every five samples is
saved in the posterior sample. The SEED= option specifies a seed for the random number generator (the seed
guarantees the reproducibility of the random stream). The PROPCOV=QUANEW option uses the estimated
inverse Hessian matrix as the initial proposal covariance matrix. The MONITOR= option specifies a list of
symbols, which can be either parameters or functions of the parameters in the model, for which inference is
to be done. The symbol _parms_ is a shorthand for all model parameters, in this case, beta0, beta1, beta2,
beta3, sig2. The symbol ExpBeta2 and ExpBeta3 are defined in the program for eˇ2 and eˇ3 , respectively.

ods graphics on;
proc mcmc data=brainweight nbi=5000 nmc=25000 thin=5 seed=1181

propcov=quanew monitor=(_parms_ ExpBeta2 ExpBeta3);

array mu_pr[4] mu_pr0-mu_pr3;
array beta[4] beta0-beta3;
array data[4] 1 log_body log_gestation litter;

begincnst;
call zeromatrix(mu_pr);
rc = logmpdfset('lp',16,0,16,0,3,2.25,0,-2,0,1);

endcnst;

parms beta: 0;
parms sig2 1;

ExpBeta2=exp(beta2);
ExpBeta3=exp(beta3);

lmvn = logmpdfnormal(of beta0-beta3, of mu_pr0-mu_pr3, 'lp');
prior beta: ~ general(lmvn);
prior sig2 ~ igamma(shape = 2.001, scale = 1.001);

call mult(beta, data, mu);
model log_brain ~ n(mu, var = sig2);

run;

data _null_;
rc = logmpdffree();
put rc=;

run;
ods graphics off;

Each of the two ARRAY statements associates a name with a list of variables and constants. The first
ARRAY statement is used to specify the prior mean. The second ARRAY statement specifies names for
the regression coefficients. The third ARRAY statement contains all of the covariates and is used to take
advantage of PROC MCMC’s ability to use matrix functions that make the code succinct.

The BEGINCNST and ENDCNST statements are used in conjunction with the DATA step functions. Pro-
gramming statements enclosed by the BEGINCNST and ENDCNST statements are executed once per pro-
cedure call. You use the LOGMPDFSET function within the statements to set up a symmetric positive
definite matrix from its lower triangular elements listed in row-major format. The RC assignment statement
is set to 0 when the numeric arguments describe a positive definite matrix; otherwise it is set to a nonzero
value.

The first PARMS statement places all regression parameters in a single block and assigns them an initial
value of 0. The second PARMS statement places the variance parameter in a separate block and assigns it
an initial value of 1.

ExpBeta1 and ExpBeta2 are the names for the functions of parameters for eˇ2 and eˇ3 that you want to
obtain posterior summary statistics and convergence diagnostics graphs.

The LMVN assignment statement uses the LOGMPDFNORMAL function to calculate the log density of
the multivariate normal density given in Equation 13. Next, the PRIOR statement for the ˇ parameters uses
the GENERAL function, which indicates that you are using a SAS statement to construct a nonstandard
density or distribution. The argument is an expression that takes the value of the logarithm function of the
prior or likelihood distribution. In this case, the nonstandard density is the multivariate normal distribution

and the argument is the logarithm of its density. The second PRIOR statement assigns the inverse-gamma
prior distribution to �2 as given in Equation 14.

The CALL statement uses the MULT matrix multiplication subroutine to calculate �i . The MODEL state-
ment specifies the likelihood function as given in Equation 12.

The DATA step after the call to the MCMC procedure calls the LOGMPDFFREE function. Use the
LOGMPFFREE function to free the workspace previously allocated with the LOGMPDFSET function.
When called without arguments, the LOGMPDFFREE frees all the symbols previously allocated by LOGM-
PDFSETSQ or LOGMPDFSET. Each freed symbol is reported in the SAS log.

Figure 9 displays convergence diagnostic graphs to assess whether the Markov chain has converged. The
trace plot indicates that the chain appears to have reached a stationary distribution. The chain also has good
mixing and is dense.

Figure 9 Diagnostic Plots for ˇ1

The autocorrelation plot indicates low autocorrelation and efficient sampling. Finally, the kernel density
plot shows the smooth, unimodal shape of posterior marginal distribution for ˇ0. In a similar fashion, the
rest of the diagnostic plots should be examined to ensure convergence of all parameters.

Figure 10 Posterior Model Summary of Bayesian MLR

The MCMC Procedure

Posterior Summaries

Standard Percentiles
Parameter N Mean Deviation 25% 50% 75%

beta0 5000 0.7583 0.6518 0.3168 0.7525 1.1771
beta1 5000 0.5703 0.0333 0.5474 0.5688 0.5924
beta2 5000 0.4527 0.1352 0.3654 0.4546 0.5445
beta3 5000 -0.1074 0.0427 -0.1363 -0.1074 -0.0780
sig2 5000 0.2434 0.0354 0.2185 0.2405 0.2649

Posterior Intervals

Parameter Alpha Equal-Tail Interval HPD Interval

beta0 0.050 -0.4789 2.0902 -0.4861 2.0695
beta1 0.050 0.5072 0.6371 0.5064 0.6356
beta2 0.050 0.1744 0.7064 0.1890 0.7180
beta3 0.050 -0.1934 -0.0244 -0.1898 -0.0226
sig2 0.050 0.1840 0.3227 0.1793 0.3157

Figure 10 reports summary and interval statistics for each regression parameter and the variance parameter.
Note that for the slope coefficient ˇ0 both equal tail interval and HPD interval contain 0.

Figure 11 Diagnostic Plots for eˇ2 and eˇ3

Figure 11 displays convergence diagnostic graphs for eˇ2 and eˇ3 . The brain weight increases by a factor

of eˇ2 D 1:5365 (approximately 54%) for each logarithmic change of one day of gestational age. Similarly,
the brain weight decreases by 1 � eˇ3 D 0:11 (11%) for each addition to the litter size.

